

9/3.5

Awari

Ce jeu de réflexion se joue contre l'ordinateur. Le principe du jeu est simple : un circuit sépare l'écran en deux camps. Le joueur se trouve en bas, et l'ordinateur en haut. Les nombres présents sur les côtés indiquent les scores joueur (à droite) et ordinateur (à gauche). Au départ, toutes les cases sont à 3. Grâce aux touches-flèches, choisissez un des nombres de votre camp, puis appuyez sur la lettre « T ». Cette case passe à 0. Si son contenu était n, les n cases situées en aval (dans une rotation de sens inverse des aiguilles d'une montre) sont incrémentées de 1.

Pour gagner, il faut :

- annuler toutes les cases de son camp,
- avoir une case total supérieure à celle de l'ordinateur.

```

1000 '=====
1010 ' Presentation du jeu
1020 '=====
1030 MODE 2:PRINT"AWARI"
1040 PRINT:PRINT"Le but du jeu est d'annuler le nombre de points dans son camp"
1050 PRINT"en tenant compte des regles suivantes:"
1060 PRINT"1) Jouer une case qui contient n pions provoque n additions"
1070 PRINT"   d'un pion sur les n cases situees apres cette case (dans le"
1080 PRINT"   sens inverse des aiguilles d'une montre)"
1090 PRINT"2) Si la derniere addition tombe sur une case total, a droite ou a ga
uche,"
1100 PRINT"le joueur rejoue"
1110 PRINT:PRINT"Appuyez sur une touche quelconque pour commencer a jouer"
1120 A#=INKEY#:IF A#="" THEN 1120
1130 '-----
1140 'Initialisation du jeu
1150 '-----
1160 'Affichage du jeu
1170 '-----
1180 MODE 1
1190 FOR I=0 TO 1
1200   FOR J=0 TO 5
1210     LOCATE 9+J*4,12+I*6:PRINT"3"
1220   NEXT J
1230 NEXT I
1240 LOCATE 5,15:PRINT"0":LOCATE 33,15:PRINT"0"
1250 '
1260 DIM T(14)
1270 FOR I=1 TO 14
1280   T(I)=3
1290 NEXT I
1300 T(1)=0:T(8)=0

```

```

1310 '-----
1320 'Deroulement du jeu
1330 '-----
1340 GOSUB 1430 'Joueur
1350 IF FIN=1 THEN GOSUB 2410
1360 IF BIS=1 THEN 1340
1370 GOSUB 1880 'Amstrad
1380 IF BIS=1 THEN 1370
1390 IF FIN=0 THEN 1320
1400 GOSUB 2410
1410 END
1420 '-----
1430 'Jeu du joueur
1440 '-----
1450 LOCATE 9,20:PRINT CHR$(196):PX=0:BIS=0:SO=0
1460 LOCATE 1,5:PRINT"A vous de jouer"
1470 LOCATE 1,22:PRINT"Servez-vous des touches fleches droite"
1480 PRINT"et gauche pour positionner la case"
1490 PRINT"desiree. Appuyez sur T quand vous "
1500 PRINT"serez positionne."
1510 B#=INKEY#:IF B#="" THEN 1510
1520 B=ASC(B#)
1530 IF B=243 THEN GOSUB 1590 'Vers la droite
1540 IF B=242 THEN GOSUB 1670 'Vers la gauche
1550 IF B#="T" THEN GOSUB 1760
1560 IF B#="T" THEN RETURN
1570 GOTO 1510
1580 '.....
1590 'Deplacement vers la droite
1600 '.....
1610 IF PX=5 THEN RETURN
1620 LOCATE 9+PX*4,20:PRINT" "
1630 PX=PX+1
1640 LOCATE 9+PX*4,20:PRINT CHR$(196)
1650 B=0
1660 RETURN :
1670 '.....
1680 'Deplacement vers la gauche
1690 '.....
1700 IF PX=0 THEN RETURN
1710 LOCATE 9+PX*4,20:PRINT" "
1720 PX=PX-1
1730 LOCATE 9+PX*4,20:PRINT CHR$(196)
1740 B=0
1750 RETURN
1760 '
1770 LOCATE 1,5:PRINT SPACE$(20)
1780 LOCATE 1,20:PRINT SPACE$(235)
1790 '.....
1800 'Calcul et affichage du coup joue
1810 '.....
1820 JEU=0:GOSUB 2140:SO=0
1830 FOR I=2 TO 7
1840   SO=SO+T(I)
1850 NEXT I
1860 IF SO=0 THEN FIN=1:GA=1
1870 RETURN
1880 '-----
1890 'Jeu Amstrad
1900 '-----
1910 MX=-100:BIS=0:SO=0
1920 '

```

```

1930 FOR I=9 TO 14
1940   C=0:D=0
1950   IF T(I)=0 THEN 2020
1960   FOR J=1 TO T(I)
1970     D=I+J
1980     IF D<15 AND D>8 THEN C=C-1
1990     IF D>1 AND D<8 THEN C=C+1
2000   NEXT J
2010   IF C>MX THEN MX=C:CASE=I
2020 NEXT I
2030 PX=CASE-2
2040 '
2050 LOCATE 1,5:PRINT"Je joue en "
2060 JEU=1:GOSUB 2140
2070 LOCATE 1,5:PRINT SPACE$(20):SO=0
2080 '
2090 FOR I=9 TO 14
2100   SO=SO+T(I)
2110 NEXT I
2120 IF SO=0 THEN FIN=1:GA=0
2130 RETURN
2140 '.....
2150 'Calcul et affichage du coup joue
2160 '.....
2170 A=PX+2:IF T(A)=0 THEN RETURN
2180 FOR I=1 TO T(A)
2190   IF A+I<=14 THEN T(A+I)=T(A+I)+1
2200   IF A+I>14 THEN T(A+I-14)=T(A+I-14)+1
2210 NEXT I
2220 IF A+I=9 OR A+I=16 THEN BIS=1
2230 T(A)=0
2240 IF JEU=1 THEN LOCATE 9+(14-A)*4,10:PRINT CHR$(198):FOR I=1 TO 500:NEXT
2250 '
2260 FOR I=0 TO 5
2270   LOCATE 8+I*4,12:PRINT T(14-I);
2280 NEXT I
2290 '
2300 FOR I=0 TO 5
2310   LOCATE 8+I*4,18:PRINT T(I+2);
2320 NEXT I
2330 '
2340 LOCATE 4,15:PRINT T(1)
2350 '
2360 B=INT(T(8)/10)
2370 LOCATE 32,15:PRINT T(8)
2380 '
2390 IF JEU=1 THEN LOCATE 9+(14-A)*4,10:PRINT" "
2400 RETURN
2410 '-----
2420 'Partie terminee
2430 '-----
2440 CLS
2450 PRINT"Partie terminee"
2460 PRINT:PRINT:PRINT
2470 IF GA=1 THEN PRINT"Bravo, vous avez gagne " ELSE PRINT"Vqus avez perdu "
2480 IF T(1)>T(8) THEN PRINT "et vous etes batu sur le nombre de points "; ELSE
PRINT"et vous gagnez au nombre de points ";
2490 PRINT "(";T(1);"contre";T(8);")"

```

Lignes 1000 à 1120 : Présentation du jeu et rappel des règles

Lignes 1150 à 1300 : Affichage du parcours initial

Lignes 1310 à 1410 : Programme principal qui oriente vers les sous-programmes joueur et ordinateur

Lignes 1420 à 1570 : Action du joueur

Lignes 1580 à 1660 : Déplacement du curseur vers la droite

Lignes 1670 à 1780 : Déplacement du curseur vers la gauche

Lignes 1790 à 1870 : Détermine si la partie est gagnée

Lignes 1880 à 2130 : Jeu de l'ordinateur

Lignes 2140 à 2400 : Répercussions du jeu de l'ordinateur sur le parcours et calcul si la partie est terminée

Lignes 2410 à 2490 : Commentaires de fin de partie